



**ESKİŞEHİR OSMANGAZİ ÜNİVERSİTESİ**

Mühendislik Mimarlık Fakültesi

İnşaat Mühendisliği Bölümü

E-Posta: [ogu.ahmet.topcu@gmail.com](mailto:ogu.ahmet.topcu@gmail.com)

Web: <http://mmf2.ogu.edu.tr/atopcu>

**Bilgisayar Destekli**

# Nümerik Analiz

*Ders notları 2014*

**Ahmet TOPÇU**

```
C:\Basic\QBBasic.EXE
File Edit View Search Run Debug Calls Utility Options Help
PCGSPAR~1.BAS:PCGsparse
DEFINT I-N
DEFDBL A-H, O-Z
SUB PCGsparse (n, ites, iA(), jA(), a(), b(), x(), iTer, iHata)
'-----
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 2010
' a(n,n)*x(n)=b(n) doğrusal denklem sistemi PCGsparse (Preconditioned
' Conjugate Gradient) iterasyon metodu ile çözülerek x(n) vektörü
' hesaplanır. a matrisinin kondisyonu çözüm öncesi iyileştirilir.
' a(n,n): simetrik pozitif tanımlı seyrek(sparse) matris
' b(n): karşı taraf vektörü
' x(n): bilinmeyenler vektörü
'-----
' Veriler:
' n: Denklem sayısı
' ites: a n'n sıfırdan farklı eleman sayısı(simetri dikkate alınmaz)
' iA(ites): a n'n sıfırdan farklı olan a(i,j) elemanlarının i satır
' numaralarının depolandığı vektör
' jA(ites): a n'n sıfırdan farklı olan a(i,j) elemanlarının j satır
' numaralarının depolandığı vektör
' a(ites) seyrek matrisinin sıfırdan farklı tüm elemanları(simetri
' dikkate alınmaz, tüm elemanlar verilir)
' b(n): karşı taraf vektörü
'-----
' Çıktılar:
' x(n): bilinmeyenler vektörü
' iTer: iterasyon sayısı
' iHata:
' = 0: Çözüm bulundu
' = 1: Çözüm bulunamadı(a tekildir veya pozitif tanımlı değildir)
' = 2: n+2 iterasyon yapıldı fakat çözüm yakınsamadı
'-----
' d(n), r(n), s(n) ve aSi(n) ara değer vektörleridir
'-----
' Çağrılan alt programlar: yok
'-----
DIM d(n) 'a n'n diyagonal elemanları
DIM r(n) 'iterasyon başlangıç vektörü
DIM s(n) 'iterasyon başlangıç vektörü
DIM aSi(n) ' a * s çarpımı
'-----
' EpsMach: Kullanılan bilgisayarda sıfır varsayılan en küçük sayı
EpsMach = 1
```

# 24

**PROGRAMLAR:**

**İterasyon ile seyrek denklem sistemi çözümü  
PCGsparse (Preconditioned Conjugate Gradient) metodu**

## 24. PROGRAMLAR: iterasyon ile seyrek katsayılı denklem sistemi çözümü- PCGsparse (Preconditioned Conjugate Gradient) metodu

Ekte verilen program katsayılar matrisi simetrik ve seyrek olan  $\underline{A} \underline{x} = \underline{b}$  doğrusal cebirsel denklem sistemini PCG iterasyon yöntemi ile çözer. Çözüm öncesi  $\underline{A}$  seyrek matrisinin kondisyonu iyileştirilir. İyileştirme  $\underline{A}$  nın diyagonal elemanlarından oluşan  $\underline{D}$  diyagonal matrisi yardımıyla yapılır:

$$\underline{A} \rightarrow \text{Kondisyon iyileştirmesi} \rightarrow \underline{D}^{-1/2} \underline{A} \underline{D}^{-1/2} \underline{D}^{1/2} \underline{x} = \underline{D}^{-1/2} \underline{b}$$

$\underline{A}_{n \times n}$  simetrik-pozitif tanımlı matrisi ve  $\underline{b}_n$  vektörü çağıran programda aşağıdaki gibi depolanmış olmalıdır:

$$\underline{a} = \begin{bmatrix} a_{11} & 0 & a_{13} & a_{14} & 0 \\ 0 & a_{22} & 0 & 0 & a_{25} \\ a_{31} & 0 & a_{33} & a_{34} & 0 \\ 0 & a_{42} & 0 & a_{44} & 0 \\ 0 & a_{52} & 0 & 0 & a_{55} \end{bmatrix}$$

Sevrek matris

Simetri dikkate alınmaz!

$a \rightarrow [a_{11} \ a_{13} \ a_{14} \ a_{22} \ a_{25} \ a_{31} \ a_{33} \ a_{34} \ a_{42} \ a_{44} \ a_{52} \ a_{55}]$

Sadece sıfırdan farklı tüm elemanlar tek boyutlu  $a$  vektöründe depolanmış olmalı

$ia \rightarrow [1 \ 1 \ 1 \ 2 \ 2 \ 3 \ 3 \ 3 \ 4 \ 4 \ 5 \ 5]$

Sıfırdan farklı elemanların satır numaraları  $ia$  vektörde depolanmış olmalı

$ja \rightarrow [1 \ 3 \ 4 \ 2 \ 5 \ 1 \ 3 \ 4 \ 2 \ 4 \ 2 \ 5]$

Sıfırdan farklı elemanların kolon numaraları  $ja$  vektörde depolanmış olmalı

PCG metodu sadece simetrik ve pozitif tanımlı matrisler içindir. Bu koşulu sağlamayan denklem sistemlerinde kullanılmaz! Maksimum iterasyon sayısı, teorik olarak,  $n$  kadardır. Ancak, yuvarlama hataları nedeniyle fazladan 1-2 iterasyon daha gerekebilir.

İterasyon sırasında  $\underline{A}$  nın pozitif tanımlı olmadığı anlaşılırsa veya  $n+2$  iterasyon sonunda çözüm yakınsamazsa PCG alt programı  $iHata \neq 0$  ile döner.

$\underline{A}$  simetrik ve pozitif tanımlı olmak kaydıyla, PCG metodu en uygun iterasyon metodudur. İterasyon sırasında  $\underline{A}$  nın elemanları değişmediğinden, bellekte çözümü mümkün olmayan çok büyük denklem sistemlerinin çözümünde öncelikle tercih edilir.

### Örnek:

$$\underline{A} \underline{x} = \underline{b} \rightarrow \begin{bmatrix} 33 & 0 & -10 & 0 & 0 \\ 0 & 43 & 0 & 14 & 0 \\ -10 & 0 & 28 & 0 & 1 \\ 0 & 14 & 0 & 55 & 0 \\ 0 & 0 & 1 & 0 & 67 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \\ x_5 \end{bmatrix} = \begin{bmatrix} 3 \\ 13 \\ 76 \\ 206 \\ 137 \end{bmatrix}, \underline{x} = ?$$

Seyrek ve simetrik matris

### Verilerin hazırlanması:

Denklem satısı  $n=5$

$\underline{A}$  nın Sıfırdan farklı elman sayısı  $iTes=11$

$\underline{A}$  nın depolanma şekli:

$a \rightarrow [33 \ -10 \ 43 \ 14 \ -10 \ 28 \ 1 \ 14 \ 55 \ 1 \ 67]$

A nın sıfırdan farklı tüm elemanları(simetri dikkate alınmaz!)

$ia \rightarrow [1 \ 1 \ 2 \ 2 \ 3 \ 3 \ 3 \ 4 \ 4 \ 5 \ 5]$

Sıfırdan farklı elemanların satır numaraları

$ja \rightarrow [1 \ 3 \ 2 \ 4 \ 1 \ 3 \ 5 \ 2 \ 4 \ 3 \ 5]$

Sıfırdan farklı elemanların kolon numaraları

$b \rightarrow [3 \ 13 \ 76 \ 206 \ 137]$

Karşı taraf vektörü

```

C:\Basic\QBASIC.EXE
Denklem sisteminin çözümü(PCGsparse):
1 -1 3 4 2
İterasyon sayısı= 6

```

$$\underline{x} = \begin{bmatrix} 1 \\ -1 \\ 3 \\ 4 \\ 2 \end{bmatrix}$$

PCGsparse alt programının sonucu

```

'----- Ana program PCGsparse(Preconditioned Conjugate Gradient Metodu)---
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 2010
' a(n,n)*x(n)=b(n) doğrusal denklem sistemi PCGsparse (Preconditioned
' Conjugate Gradient) iterasyon metodu ile çözümlenerek x(n) vektörü
' hesaplanır.

```

PCG ana programı

```

' a(n,n): simetrik pozitif tanımlı seyrek(sparse) matris
' b(n): karşı taraf vektörü
' x(n): bilinmeyenler vektörü
' n: Denklem sayısı

```

```

' a(n,n) seyrek matrisinin sadece sıfırdan farklı sayıları
' a(iTes) bir boyutlu vektöründe depolanır
' sıfırdan farklı elemanların satır numaraları iA(iTes), kolon numaraları
' jA(iTes) vektörlerinde depolanır

```

```

' iTes: a nın sıfırdan farklı eleman sayısıdır.
' iA(ites): a nın sıfırdan farklı olan a(i,j) elemanlarının i satır
' numaralarının depolandığı vektördür
' jA(ites): a nın sıfırdan farklı olan a(i,j) elemanlarının j satır
' numaralarının depolandığı vektördür

```

```

' DATA satırları iA, jA, a(i,j) sırasında düzenlenmiştir

```

```

' Çağrılan alt programlar: PCGsparse

```

```

-----
DATA 5: ' denklem sayısı n
DATA 11: ' A nın sıfırdan farklı sayı sayısı iTes
' A nın sıfırdan farklı sayıları: iA, jA, a(i,j)
DATA 1,1,33, 1,3,-10
DATA 2,2,43, 2,4,14
DATA 3,1,-10, 3,3,28, 3,5,1
DATA 4,2,14, 4,4,55
DATA 5,3,1, 5,5,67
' b karşı taraf vektörü:
DATA 3,13,76,206,137

```

i, j, a(i,j) değerleri

```

DEFINT I-N
DEFDBL A-H, O-Z
DECLARE SUB PCGsparse (n, ites, iA(), jA(), a(), b(), x(), iTer, iHata)

```

```
CLS
```

Denklem sayısı okunuyor

```
READ n
```

Sıfırdan farklı eleman sayısı

```
READ ites
```

```
DIM a(ites), iA(ites), jA(ites), x(n), b(n)
```

A, iA ve ja için yer ayır

```
' a nın elemanlarını oku
```

```
FOR i = 1 TO ites
```

```
  READ iA(i), jA(i), a(i)
```

```
NEXT i
```

iA, jA ve a değerleri okunuyor

```
' b yi oku
```

```
FOR i = 1 TO n
```

```
  READ b(i)
```

```
NEXT i
```

Karşı taraf vektörü okunuyor

```
CALL PCGsparse(n, ites, iA(), jA(), a(), b(), x(), iTer, iHata)
```

PCGsparse alt programı çağrılıyor

```
IF iHata <> 0 THEN PRINT "HATA(PCGsparse)="; iHata: END
```

```
PRINT "Denklem sisteminin çözümü(PCGsparse):"
```

```
FOR i = 1 TO n
```

```
  PRINT x(i);
```

```
NEXT i
```

```
PRINT
```

```
PRINT "İterasyon sayısı="; iTer
```

Çözüm yazdırılıyor

```
END ' PCGsparse ana
```

```

SUB PCGsparse (n, ites, iA(), jA(), a(), b(), x(), iTer, iHata)
-----
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 2010
' a(n,n)*x(n)=b(n) doğrusal denklem sistemi PCGsparse
' (Preconditioned
' Conjugate Gradient) iterasyon metodu ile çözülerek x(n) vektörü
' hesaplanır. a matrisinin kondisyonu çözüm öncesi iyileştirilir.
' a(n,n): simetrik pozitif tanımlı seyrek(sparse) matris
' b(n): karşı taraf vektörü
' x(n): bilinmeyenler vektörü

' Veriler:
' n: Denklem sayısı
' iTes: a n'ın sıfırdan farklı eleman sayısı(simetri dikkate alınmaz)
' iA(it): a n'ın sıfırdan farklı olan a(i,j) elemanlarının i satır
' numaralarının depolandığı vektör
' jA(it): a n'ın sıfırdan farklı olan a(i,j) elemanlarının j satır
' numaralarının depolandığı vektör
' a(iT) seyrek matrisinin sıfırdan farklı tüm elemanları(simetri
' dikkate alınmaz, tüm elemanlar verilir)
' b(n): karşı taraf vektörü

' Çıktılar:
' x(n): bilinmeyenler vektörü
' iTer: iterasyon sayısı
' iHata:
' = 0: Çözüm bulundu
' = 1: Çözüm bulunamadı(a tekdirdir veya pozitif tanımlı değildir)
' = 2: n+2 iterasyon yapıldı fakat çözüm yakınsamadı

' d(n), r(n), s(n) ve aSi(n) ara değer vektörleridir
' Çağrılan alt programlar: yok
-----
DIM d(n) 'a n'ın diyagonal elemanları
DIM r(n) 'iterasyon başlangıç vektörü
DIM s(n) 'iterasyon başlangıç vektörü
DIM aSi(n) 'a * s çarpımı

' EpsMach: Kullanılan bilgisayarda sıfır varsayılan en küçük sayı
EpsMach = 1
DO
  EpsMach = EpsMach / 2
  s = 1 + EpsMach
LOOP UNTIL s <= 1
EpsMach = 2 * EpsMach
epsilon = EpsMach

iHata = 0
iTer = 0
' n=1 durumu
IF n = 1 THEN
  IF a(1) = 0 THEN
    iHata = 1
  ELSE
    x(1) = b(1) / a(1)
    EXIT SUB
  END IF
END IF

'kondisyonu iyileştir(Jacobi)
FOR i = 1 TO ites
  iA = iA(i)
  IF iA = jA(i) THEN
    d(iA) = SQR(a(i))
    b(iA) = d(iA) * b(iA)
  END IF
NEXT i
FOR i = 1 TO ites
  iA = iA(i)
  jA = jA(i)
  a(i) = d(iA) * a(i) * d(jA)
NEXT i

' x, r, s başlangıç vektörlerini hazırla
FOR i = 1 TO n
  x(i) = 0
  r(i) = b(i)
  s(i) = r(i)
NEXT i

' B nin öklid normu
bNorm = 0
FOR i = 1 TO n
  bNorm = bNorm + b(i) * b(i)
NEXT i
IF bNorm = 0 THEN EXIT SUB
bNorm = SQR(bNorm)

```

**PCGsparse alt programı**

```

' İterasyona başla
iTer = 1
DO
' Hesapla: Alfa = (s(transpoz) * r) / (s(transp) * (a * s))
pay = 0
FOR i = 1 TO n
  pay = pay + s(i) * r(i)
NEXT i

' hesapla: a*s
FOR i = 1 TO n
  aSi(i) = 0
NEXT i

FOR i = 1 TO ites
  aSi(iA(i)) = aSi(iA(i)) + a(i) * s(jA(i))
NEXT i

' Hesapla s(transpoz)*(aSi)
payda = 0
FOR i = 1 TO n
  payda = payda + s(i) * aSi(i)
NEXT i

' a matrisi tekil mi? Veya pozitif tanımlı değil mi? Kontrol et
IF payda <= 0 THEN
  iHata = 1
  EXIT SUB
END IF
alfa = pay / payda

' Hesapla: x=x+Alfa*s ve r=r-Alfa*aSi
' Hesapla: rNorm=r nin öklid normu
rNorm = 0
FOR i = 1 TO n
  x(i) = x(i) + alfa * s(i)
  r(i) = r(i) - alfa * aSi(i)
  rNorm = rNorm + r(i) * r(i)
NEXT i
rNorm = SQR(rNorm) / bNorm

' Yakınsamayı kontrol et (en az iki iterasyondan sonra)
IF iTer > 1 AND rNorm <= epsilon THEN
  FOR i = 1 TO n
    x(i) = d(i) * x(i)
  NEXT i
  EXIT SUB
END IF

' Hesapla: Beta = -[r(transpoz)*(a*s)]/[s(transpoz)*(a * s)]
pay = 0
FOR i = 1 TO n
  pay = pay + r(i) * aSi(i)
NEXT i
beta = -pay / payda

' Hesapla: s = r + Beta * s
FOR i = 1 TO n
  s(i) = r(i) + beta * s(i)
NEXT i

  iTer = iTer + 1
LOOP UNTIL iTer > n + 2

iHata = 2
END SUB ' PCGsparse sonu

```